

## Chapter 13 Calculus of Vector-Valued Functions

**Useful Tip:** If you are reading the electronic version of this publication formatted as a *Mathematica* Notebook, then it is possible to view 3-D plots generated by *Mathematica* from different perspectives. First, place your screen cursor over the plot. Then drag the mouse while pressing down on the left mouse button to rotate the plot.

### ■ 13.1. Vector-Valued Functions

**Students should read Section 13.1 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.**

A vector-valued function is a vector where the components of the vector are themselves functions of a common parameter (or variable). For example,  $\mathbf{r}$  is a vector-valued function if  $\mathbf{r}(t) = \langle x(t), y(t), z(t) \rangle$ . If we think of  $t$  as the time variable, the  $\mathbf{r}(t)$  describes the motion of a particle through three-dimensional space over time. What we want to do is to understand what path is taken. We do this through graphing in three dimensions. Also, sometimes it is helpful to consider the projections of these curves onto the coordinate planes. For example, the projection of  $\mathbf{r}(t)$  on the  $xy$ -plane is  $\langle x(t), y(t), 0 \rangle$ .

**Example 13.1.** Trace the paths of each of the following vector functions and describe its projections onto the  $xy$ -,  $xz$ -, and  $yz$ -planes:


a)  $\mathbf{r}(t) = \langle t, t^2, 2t \rangle$

b)  $\mathbf{r}(t) = \langle \cos^3 t, \sin^3 t, \sin 2t \rangle$

**Solution:** We use the **ParametricPlot3D** command to trace the path of each curve and to see its projection.


a) First, we look at the plot of  $\mathbf{r}(t) = \langle t, t^2, 2t \rangle$ :

```
In[73]:= ParametricPlot3D[{t, t^2, 2t}, {t, -3, 3}, PlotStyle -> Red, ImageSize -> {250}]
```




This curve looks very much like a parabola in 3-D space. To see the projections, we look first at:

```
In[74]:= ParametricPlot3D[{t, t^2, 0}, {t, -3, 3},
  PlotRange -> {-1, 1}, PlotStyle -> Orange, ImageSize -> {250}]
```




This is clearly a parabola in the xy-plane.

```
In[75]:= ParametricPlot3D[{t, 0, 2 t}, {t, -3, 3}, Ticks -> {Automatic, {-1, 0, 1}, Automatic},
  PlotStyle -> Orange, ImageSize -> {250}, ImagePadding -> {{15, 15}, {15, 15}}]
```



And this clearly a line in the xz-plane.


```
In[76]:= ParametricPlot3D[{0, t^2, 2 t}, {t, -3, 3},
  Ticks -> {{-1, 0, 1}, Automatic, Automatic}, PlotStyle -> Orange, ImageSize -> {250}]
```



This last plot is also clearly a parabola, but in the  $yz$ -plane.


b) Next, we look at  $\mathbf{r}(t) = \langle \cos^3 t, \sin^3 t, \sin 2t \rangle$ :

```
In[77]:= ParametricPlot3D[{Cos[t]^3, Sin[t]^3, Sin[2 t]},
  {t, -2 π, 2 π}, PlotStyle -> Orange, ImageSize -> {250}]
```




Note that since both sine and cosine are periodic with period  $2\pi$ , it is not necessary to extend the domain beyond  $-2\pi$  or  $+2\pi$ . The projection in the  $xy$ -plane is:

```
In[78]:= ParametricPlot3D[{Cos[t]^3, Sin[t]^3, 0},
  {t, -2 π, 2 π}, PlotPoints → 100, ImageSize → {250}]
```




The projection in the  $xz$ -plane is:

```
In[79]:= ParametricPlot3D[{Cos[t]^3, 0, Sin[2 t]}, {t, -2 π, 2 π}, ImageSize → {250}]
```



Lastly, the projection in the  $yz$ -plane is:

```
In[80]:= ParametricPlot3D[{0, Sin[t]^3, Sin[2 t]}, {t, -2 π, 2 π}, ImageSize -> {250}]
```



Note that the last two projections are almost exactly alike. This is to be expected because the sine and cosine functions have the same graph, but  $\pi/2$  radians apart.

## ■ Exercises

In Exercises 1 through 3, graph  $\mathbf{r}(t)$  and its three projections onto the coordinate planes.

1.  $\mathbf{r}(t) = \langle \cos(2t), \cos t, \sin t \rangle$
2.  $\mathbf{r}(t) = \langle t + 15, e^{0.08t} \cos t, e^{0.08t} \sin t \rangle$
3.  $\mathbf{r}(t) = \langle t, t, 25t/(1+t^2) \rangle$

4. Which of the following curves have the same projection onto the  $xz$ -plane? Graph the three projections to check your answer.

- a.  $\mathbf{r}_1(t) = \langle t, e^t, t^2 \rangle$
- b.  $\mathbf{r}_2(t) = \langle e^t, t, t^2 \rangle$
- c.  $\mathbf{r}_3(t) = \langle t, \cos t, t^2 \rangle$

## ■ 13.2. Calculus of Vector-Valued Functions

**Students should read Section 13.2 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.**

Since vector-valued functions are differentiated and integrated component by component, *Mathematica* will handle this easily since it treats vectors as lists and automatically performs the indicated operation on each element of the list.

The derivative of a vector valued function  $\mathbf{r}(t) = \langle x(t), y(t), z(t) \rangle$  is defined to be

$$\mathbf{r}'(t) = \langle x'(t), y'(t), z'(t) \rangle$$

while the integral of  $\mathbf{r}(t)$  is

$$\int \mathbf{r}(t) dt = \left\langle \int x(t) dt, \int y(t) dt, \int z(t) dt \right\rangle.$$

Similarly, the limit is defined by

$$\lim_{t \rightarrow a} \mathbf{r}(t) = \langle \lim_{t \rightarrow a} x(t), \lim_{t \rightarrow a} y(t), \lim_{t \rightarrow a} z(t) \rangle.$$

**Example 13.2.** Differentiate and integrate each of the following vector functions:

a)  $\mathbf{r}(t) = \langle t, t^2, 2t \rangle$

b)  $\mathbf{s}(t) = \langle \cos^3 t, \sin^3 t, \sin 2t \rangle$

**Solution:**

(a)

In[81]:= `Clear[r, s, t]`

In[82]:= `r[t_] := {t, t^2, 2 t}`  
`s[t_] := {Cos[t]^3, Sin[t]^3, Sin[2 t]}`

In[84]:= `D_t r[t]`

Out[84]= `{1, 2 t, 2}`

In[85]:= `Integrate[r[t], t]`

Out[85]= `{t^2/2, t^3/3, t^2}`

(b)

In[86]:= `D_t s[t]`

Out[86]= `{-3 Cos[t]^2 Sin[t], 3 Cos[t] Sin[t]^2, 2 Cos[2 t]}`

In[87]:= `Integrate[s[t], t]`

Out[87]= `{3 Sin[t]/4 + 1/12 Sin[3 t], -3 Cos[t]/4 + 1/12 Cos[3 t], -1/2 Cos[2 t]}`

Limits are handled the same way both in the calculus of vector-valued functions and in *Mathematica*:

**Example 13.3.** Evaluate  $\lim_{h \rightarrow 0} \frac{\mathbf{r}(t+h) - \mathbf{r}(t)}{h}$  for  $\mathbf{r}(t) = \langle t, t^2, 2t \rangle$ .

**Solution:**

Since  $\mathbf{r}(t)$  has been defined in the previous example, we merely evaluate

In[88]:= `Limit[(r[t+h] - r[t])/h, h -> 0]`

Out[88]= `{1, 2 t, 2}`

As we would expect, this limit gives us the same answer for  $\mathbf{r}'(t)$  as in the previous example.

**Example 13.4.** Evaluate  $\lim_{t \rightarrow 3} \langle t^2, 4t, \frac{1}{t^3} \rangle$ .

**Solution:**

$$\text{In[89]:= Limit}\left[\left\{t^2, 4t, \frac{1}{t^3}\right\}, t \rightarrow 2\right]$$

$$\text{Out[89]= } \left\{4, 8, \frac{1}{8}\right\}$$

### Derivatives of Dot and Cross Products

Using the formulas of the derivative of the dot and cross products for vector-valued functions is simple in *Mathematica*. As a reminder, the formulas are:

$$\frac{d}{dt}(\mathbf{r}(t) \cdot \mathbf{s}(t)) = \mathbf{r}(t) \cdot \mathbf{s}'(t) + \mathbf{r}'(t) \cdot \mathbf{s}(t) \text{ and } \frac{d}{dt}(\mathbf{r}(t) \times \mathbf{s}(t)) = \mathbf{r}(t) \times \mathbf{s}'(t) + \mathbf{r}'(t) \times \mathbf{s}(t)$$

**Example 13.5.** Evaluate  $\frac{d}{dt}(\mathbf{r}(t) \cdot \mathbf{s}(t))$  and  $\frac{d}{dt}(\mathbf{r}(t) \times \mathbf{s}(t))$  for  $\mathbf{r}(t) = \langle t, t^2, 2t \rangle$  and  $\mathbf{s}(t) = \langle \cos^3 t, \sin^3 t, \sin 2t \rangle$ .

**Solution:**

$$\text{In[90]:= } \partial_t(\mathbf{r}[t] \cdot \mathbf{s}[t])$$

$$\text{Out[90]= } \cos[t]^3 + 4t \cos[2t] - 3t \cos[t]^2 \sin[t] + 3t^2 \cos[t] \sin[t]^2 + 2t \sin[t]^3 + 2 \sin[2t]$$

$$\text{In[91]:= } \partial_t(\mathbf{r}[t] \times \mathbf{s}[t])$$

$$\begin{aligned} \text{Out[91]= } & \{2t^2 \cos[2t] - 6t \cos[t] \sin[t]^2 - 2 \sin[t]^3 + 2t \sin[2t], \\ & 2 \cos[t]^3 - 2t \cos[2t] - 6t \cos[t]^2 \sin[t] - \sin[2t], \\ & -2t \cos[t]^3 + 3t^2 \cos[t]^2 \sin[t] + 3t \cos[t] \sin[t]^2 + \sin[t]^3\} \end{aligned}$$

### Tangent Lines

**Example 13.6.** Find the vector parametrization of the tangent line to  $\mathbf{r}(t) = \langle 1 - t^2, 5t, t^3 \rangle$  at the point  $t = 1$  and plot it along with  $\mathbf{r}(t)$ .

**Solution:** Recall that the tangent line at  $t_0$  has vector parametrization  $\mathbf{L}(t) = \mathbf{r}(t_0) + t \mathbf{r}'(t_0)$ :

$$\text{In[92]:= } \mathbf{r}[t_] = \{1 - t^2, 5t, t^3\}$$

$$\mathbf{r}'[t]$$

$$\mathbf{L}[t_] = \mathbf{r}[1] + t * \mathbf{r}'[1]$$


$$\text{Out[92]= } \{1 - t^2, 5t, t^3\}$$

$$\text{Out[93]= } \{-2t, 5, 3t^2\}$$

$$\text{Out[94]= } \{-2t, 5 + 5t, 1 + 3t\}$$

Here is a plot of the curve and the tangent line.

```
In[95]:= ParametricPlot3D[{r[t], L[t]}, {t, -2, 4}, ImageSize -> Small]
```



NOTE: Recall that the plot can be rotated to better view it from different perspectives.

### ■ Exercises

In Exercises 1 and 2 evaluate the limits

$$1. \lim_{t \rightarrow \pi} \langle \sin 2t, \cos t, \tan 4t \rangle \qquad 2. \lim_{t \rightarrow 0} \left\langle \frac{1}{t+1}, \frac{e^t - 1}{t}, 4t \right\rangle$$

In Exercises 3 and 4 compute the derivative and integral.

$$3. \mathbf{r}(t) = \langle \tan t, 4t - 2, \sin t \rangle \qquad 4. \mathbf{r}(t) = \langle e^t, e^{2t} \rangle$$

5. Find a parametrization of the tangent line at the point indicated and plot both the vector-valued curve and the tangent line on the same set of axes.

6. Evaluate  $\frac{d}{dt} \mathbf{r}(g(t))$  for  $\mathbf{r}(t) = \langle 4 \sin 2t, 2 \cos 2t \rangle$  and  $g(t) = t^2$ .

## ■ 13.3. Arc Length

**Students should read Section 13.3 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.**

The arc length of a path  $\mathbf{r}(t) = \langle x(t), y(t), z(t) \rangle$  for  $a \leq t \leq b$  is given by

$$L = \int_a^b \|\mathbf{r}'(t)\| \, dt = \int_a^b \sqrt{(x'(t))^2 + (y'(t))^2 + (z'(t))^2} \, dt$$

and like the one-dimensional version is difficult to evaluate by hand. Thus *Mathematica* is the perfect tool for calculating this.

**Example 13.7.** Compute the arc length of  $\mathbf{r}(t) = \langle 1 - t^2, 5t, 2t^3 \rangle$  over the interval  $1 \leq t \leq 2$ .

**Solution:**



In[96]:=  $\mathbf{r}[t_] := \{1 - t^2, 5t, 2t^3\}$

$$L = \int_1^2 \text{Norm}[\mathbf{r}'[t]] dt$$

Out[97]= 
$$\left( \frac{1}{54} + \frac{i}{54} \right) (-1)^{3/4} \left( 9 \sqrt{130} - 18 \sqrt{1234} - \right.$$

$$(8 - 8i) \sqrt{-\frac{7}{i + 4\sqrt{14}}} \text{EllipticE}\left[ i \text{ArcSinh}\left[ \frac{3 + 3i}{\sqrt{i - 4\sqrt{14}}} \right], \frac{i - 4\sqrt{14}}{i + 4\sqrt{14}} \right] -$$

$$(1 + i) \sqrt{-\frac{2}{i + 4\sqrt{14}}} \text{EllipticE}\left[ i \text{ArcSinh}\left[ \frac{3 + 3i}{\sqrt{i - 4\sqrt{14}}} \right], \frac{i - 4\sqrt{14}}{i + 4\sqrt{14}} \right] +$$

$$(8 - 8i) \sqrt{-\frac{7}{i + 4\sqrt{14}}} \text{EllipticE}\left[ i \text{ArcSinh}\left[ \frac{6 + 6i}{\sqrt{i - 4\sqrt{14}}} \right], \frac{i - 4\sqrt{14}}{i + 4\sqrt{14}} \right] +$$

$$(1 + i) \sqrt{-\frac{2}{i + 4\sqrt{14}}} \text{EllipticE}\left[ i \text{ArcSinh}\left[ \frac{6 + 6i}{\sqrt{i - 4\sqrt{14}}} \right], \frac{i - 4\sqrt{14}}{i + 4\sqrt{14}} \right] +$$

$$(8 - 8i) \sqrt{-\frac{7}{i + 4\sqrt{14}}} \text{EllipticF}\left[ i \text{ArcSinh}\left[ \frac{3 + 3i}{\sqrt{i - 4\sqrt{14}}} \right], \frac{i - 4\sqrt{14}}{i + 4\sqrt{14}} \right] -$$

$$(224 + 224i) \sqrt{-\frac{2}{i + 4\sqrt{14}}} \text{EllipticF}\left[ i \text{ArcSinh}\left[ \frac{3 + 3i}{\sqrt{i - 4\sqrt{14}}} \right], \frac{i - 4\sqrt{14}}{i + 4\sqrt{14}} \right] -$$

$$(8 - 8i) \sqrt{-\frac{7}{i + 4\sqrt{14}}} \text{EllipticF}\left[ i \text{ArcSinh}\left[ \frac{6 + 6i}{\sqrt{i - 4\sqrt{14}}} \right], \frac{i - 4\sqrt{14}}{i + 4\sqrt{14}} \right] +$$

$$\left. (224 + 224i) \sqrt{-\frac{2}{i + 4\sqrt{14}}} \text{EllipticF}\left[ i \text{ArcSinh}\left[ \frac{6 + 6i}{\sqrt{i - 4\sqrt{14}}} \right], \frac{i - 4\sqrt{14}}{i + 4\sqrt{14}} \right] \right)$$

Note that the above output indicates that *Mathematica* cannot find an antiderivative for the integrand, and thus we need to find another technique to evaluate this integral. Hence, we next try the numerical integrate command, **NIntegrate**, which does give us our result:

In[98]:=  $L = \text{NIntegrate}[\text{Norm}[\mathbf{r}'[t]], \{t, 1, 2\}]$

Out[98]= 15.285

### Speed

The vector  $\mathbf{r}'(t)$  is also known as the *velocity vector* as it points in the (instantaneous) direction of motion described by  $\mathbf{r}(t)$ . Its length or *norm*,  $\|\mathbf{r}'(t)\|$ , gives the speed at time  $t$ .

**Example 13.8.** Compute the speed of  $\mathbf{r}(t) = \langle 1 - t^2, 5t, 2t^3 \rangle$  when  $t = 1, 1.5,$  and  $2$ .

**Solution:**

The following output gives a list of speeds of  $\mathbf{r}'(t)$  at the three given times using the **Norm** command, which calculates the norm of a vector:

```
In[99]:= r[t_] := {1 - t^2, 5 t, 2 t^3}
        Speed = {Norm[r'[1]], Norm[r'[1.5]], Norm[r'[2]]}
```

```
Out[100]= {√65, 14.7054, √617}
```


```
In[101]:= N[%]
```

```
Out[101]= {8.06226, 14.7054, 24.8395}
```

Observe that the speed is increasing as we move along the path of  $\mathbf{r}(t)$  from  $t = 1$  to  $t = 2$ . This can be seen graphically by plotting the speed:

```
In[102]:= Norm[r'[t]]
        Plot[Norm[r'[t]], {t, 1, 2}]
```

```
Out[102]= √(25 + 4 Abs[t]^2 + 36 Abs[t]^4)
```



**NOTE:** Observe how the **Norm** command inserts absolute values around each vector component in the formula for  $\|\mathbf{r}'(t)\|$ , which seems redundant since each component is squared. This is done because in *Mathematica* vector components are allowed to be complex-valued, in which case absolute values are needed to refer to their magnitudes.

## ■ Exercises

In Exercises 1 and 2, compute the length of curve over the given interval.

- $\mathbf{r}(t) = \langle 2 \sin t, 6t, 2 \cos t \rangle, \quad -6 \leq t \leq 6$
- $\mathbf{r}(t) = \langle 12t, 8t^{3/2}, 3t^2 \rangle, \quad 0 \leq t \leq 1$

In Exercises 3 and 4, find the speed of a particle moving along the curve  $\mathbf{r}(t)$  at the given value of  $t$ .

- $\mathbf{r}(t) = \langle e^{t-2}, 15t, 5/t \rangle, \quad t = 1$
- $\mathbf{r}(t) = \langle \sin 2t, \cos 4t, \sin 6t \rangle, \quad t = \pi/2$

- Compute  $s(t) = \int_0^t \|\mathbf{r}'(u)\| \, du$  for  $\mathbf{r}(t) = \langle t^2, 2t^2, t^3 \rangle$  and interpret *Mathematica's* result.

6. For  $\mathbf{r}(t) = \langle 4t, 1 - 3t, \sqrt{24}t \rangle$ , compute  $s(t)$  as in the previous exercise. Then use  $s(t)$  to find an arc length parametrization of  $\mathbf{r}(t)$ , that is, find  $\varphi(s) = t$ , where  $\varphi$  is the inverse of  $s(t)$ , and check to see that  $\mathbf{r}(\varphi(s))$  has unit speed, that is,  $\|\mathbf{r}'(\varphi(s))\| = 1$ . Lastly, plot  $\mathbf{r}(t)$  and  $\mathbf{r}(\varphi(s))$  and compare them.
7. Consider the helix  $\mathbf{r}(t) = \langle a \sin t, a \cos t, ct \rangle$ .
- Find a formula for the arc length of one revolution of  $\mathbf{r}(t)$ .
  - Suppose a helix has radius 10, height 5, and makes three revolutions. What is its arc length?
8. The Cornu spiral is defined by  $\mathbf{r}(t) = \langle x(t), y(t) \rangle$ , where  $x(t) = \int_0^t \sin\left(\frac{u^2}{2}\right) du$  and  $y(t) = \int_0^t \cos\left(\frac{u^2}{2}\right) du$ .
- Plot the Cornu spiral over various intervals for  $t$ .
  - Find a formula for its arc length along the interval  $-a \leq t \leq a$ , where  $a$  is a positive real number.
  - What is its arc length in the limit as  $a \rightarrow \infty$ ?

## ■ 13.4. Curvature

**Students should read Section 13.4 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.**

Vector tools previously studied including arc length enables one to study the idea of curvature, which serves as a measure of how a curve bends, that is, the rate of change in direction of a curve. In arriving at a definition of curvature, consider a path in vector form and parametrized by

$$\mathbf{r}(t) = \langle x(t), y(t), z(t) \rangle$$

The parametrization is classified as regular if  $\mathbf{r}'(t) \neq \mathbf{0}$  for all values of  $t$  and for which  $\mathbf{r}(t)$  is defined. Assume then that  $\mathbf{r}(t)$  is regular and define the *unit tangent vector* in the direction of  $\mathbf{r}'(t)$ , denoted  $\mathbf{T}(t)$ , as follows:

$$\mathbf{T}(t) = \frac{\mathbf{r}'(t)}{\|\mathbf{r}'(t)\|}.$$

This unit tangent vector  $\mathbf{T}$  at any point enables us to determine the direction of the curve at that point, so one may define the curvature  $\kappa$  (Greek letter kappa) at a point as

$$\kappa = \left\| \frac{d\mathbf{T}}{ds} \right\| = \frac{\|\mathbf{T}'(t)\|}{\|\mathbf{r}'(t)\|},$$

which represents the magnitude of the rate of change in the unit tangent vector with respect to arc length. One denotes the vector  $d\mathbf{T}/ds$  as the curvature vector. Its scalar length therefore measures curvature. For example, a straight line has  $\kappa = 0$  (zero curvature) as one would expect. For a circle of radius  $\rho$ , we have  $\kappa = 1/\rho$  (reciprocal of  $\rho$ ). This makes sense since a larger circle should have smaller curvature. In general, if we were to secure a circle, called the *osculating circle*, that best fits a curve at a specific point on the curve, then curvature of the curve at such a point should agree with the curvature of the osculating circle, that is,

$$\kappa = \frac{1}{\rho}$$

Moreover, the radius  $\rho$  of this circle is called the *radius of curvature*. Note that the equations linking  $\kappa$  and  $\rho$  illustrate their inverse relationship:

$$\kappa = \frac{1}{\rho} \text{ and } \rho = \frac{1}{\kappa}$$

**Example 13.9.** Compute the curvature  $\kappa$  for a circle of radius  $\rho$  defined by

$$\mathbf{r}(t) = \langle \rho \cos t, \rho \sin t \rangle$$

**Solution:** We first compute the unit tangent vector  $\mathbf{T}$  using the formula  $\mathbf{T}(t) = \frac{\mathbf{r}'(t)}{\|\mathbf{r}'(t)\|}$ :

```
In[104]:= Clear[r, T, t, ρ]
```

```
In[105]:= r[t_] = {ρ Cos[t], ρ Sin[t]}
           r'[t]
           T[t_] = r'[t] / Simplify[Norm[r'[t]]]
```

```
Out[105]= {ρ Cos[t], ρ Sin[t]}
```

```
Out[106]= {-ρ Sin[t], ρ Cos[t]}
```

```
Out[107]= { -  $\frac{\rho \sin[t]}{\sqrt{\text{Abs}[\rho \cos[t]]^2 + \text{Abs}[\rho \sin[t]]^2}}$ ,  $\frac{\rho \cos[t]}{\sqrt{\text{Abs}[\rho \cos[t]]^2 + \text{Abs}[\rho \sin[t]]^2}}$  }
```

Observe that in this output *Mathematica* is not able to reduce the expression inside the radical, which simplifies to  $\rho$  as a result of the fundamental trigonometric identity  $\cos^2 x + \sin^2 x = 1$ . This is due to the **Norm** command, which employs absolute values.

To remedy this, we use the formula  $\|\mathbf{r}'(t)\| = \sqrt{\mathbf{r}'(t) \cdot \mathbf{r}'(t)}$  instead of the **Norm** command.

```
In[108]:= T[t_] = r'[t] / Sqrt[Simplify[r'[t] . r'[t]]]
```

```
Out[108]= { -  $\frac{\rho \sin[t]}{\sqrt{\rho^2}}$ ,  $\frac{\rho \cos[t]}{\sqrt{\rho^2}}$  }
```

We then compute the curvature using the formula  $\kappa = \frac{\|\mathbf{T}'(t)\|}{\|\mathbf{r}'(t)\|}$ :

```
In[109]:= κ = Sqrt[Simplify[T'[t] . T'[t]]] / Simplify[r'[t] . r'[t]]
```

```
Out[109]=  $\sqrt{\frac{1}{\rho^2}}$ 
```

Since the radius  $\rho$  is assumed to be positive, we conclude that  $\kappa = \sqrt{\frac{1}{\rho^2}} = \left| \frac{1}{\rho} \right| = \frac{1}{\rho}$  as expected.

**Example 13.10.** Compute the curvature  $\kappa$  for the curve defined by  $f(x) = x^2$  at the point  $(3, 9)$ .

**Solution:** Observe that the graph of a function  $y = f(x)$  can be parametrized by  $x = t$  and  $y = f(t)$  and hence  $\mathbf{r}(t) = \langle t, f(t) \rangle$ . In this case the formula for curvature reduces to

```
In[110]:= Clear[r, t, f]
```

```
r[t_] = {t, f[t]}
```

```
Out[111]= {t, f[t]}
```

```
In[112]:= T[t_] =  $\frac{r'[t]}{\text{Sqrt}[r'[t].r'[t]]}$ 
 $\kappa = \text{Sqrt}[\text{Simplify}[T'[t].T'[t]] / \text{Simplify}[r'[t].r'[t]]]$ 
```

```
Out[112]=  $\left\{ \frac{1}{\sqrt{1+f'[t]^2}}, \frac{f'[t]}{\sqrt{1+f'[t]^2}} \right\}$ 
```

```
Out[113]=  $\sqrt{\frac{f''[t]^2}{(1+f'[t]^2)^3}}$ 
```

which is the same as  $\kappa = \frac{|f''(x)|}{(1+(f'(x))^2)^{3/2}}$ . With  $f(x) = x^2$ , we get

```
In[114]:= f[t_] = t^2
 $\kappa$ 
```

```
Out[114]= t^2
```

```
Out[115]=  $2 \sqrt{\frac{1}{(1+4t^2)^3}}$ 
```


At  $x = t = 3$ , the curvature becomes

```
In[116]:=  $\kappa /. t \rightarrow 3$ 
```

```
Out[116]=  $\frac{2}{37 \sqrt{37}}$ 
```

Here is a plot of the curvature along with the function.

```
In[117]:= Plot[{f[t],  $\kappa$ }, {t, 0, 3}]
```



**Example 13.10.** Compute the curvature  $\kappa$  and the radius of curvature  $\rho$  for the curve defined by

$$\mathbf{r}(t) = \left\langle 1 - t, t^2 + 1, \frac{2}{3}t^3 + 1 \right\rangle \text{ at } t = 1/2.$$

**Solution:** Again we begin by computing the unit tangent vector  $\mathbf{T}$ :

```
In[118]:= Clear[r, T, t, κ]
```

```
In[119]:= r[t_] = {1 - t, t^2 + 1, (2/3) t^3 + 1}
           r'[t]
           T[t_] = r'[t] / Sqrt[Simplify[r'[t].r'[t]]]
```

```
Out[119]= {1 - t, 1 + t^2, 1 + \frac{2 t^3}{3}}
```

```
Out[120]= {-1, 2 t, 2 t^2}
```

```
Out[121]= \left\{ -\frac{1}{\sqrt{(1 + 2 t^2)^2}}, \frac{2 t}{\sqrt{(1 + 2 t^2)^2}}, \frac{2 t^2}{\sqrt{(1 + 2 t^2)^2}} \right\}
```

We then compute the curvature using the same formula as in the previous example and evaluate it at  $t = 1/2$ :

```
In[122]:= κ = Sqrt[Simplify[T'[t].T'[t]] / Simplify[r'[t].r'[t]]]
           κ /. t -> 1/2
```

```
Out[122]= 2 \sqrt{\frac{1}{(1 + 2 t^2)^4}}
```

```
Out[123]= \frac{8}{9}
```

Hence, the curvature  $\kappa = 8/9$  at  $t = 1/2$  and the corresponding radius of curvature is  $\rho = 1/\kappa = 9/8$ .

### Curvature Formula (Cross Product)

There is an alternative formula for calculating the curvature of space curves that involves the cross product and eliminates the need to compute the unit tangent vector function:

$$\kappa = \frac{\|\mathbf{r}''(t) \times \mathbf{r}'(t)\|}{\|\mathbf{r}'(t)\|^3} = \frac{\|\mathbf{a}(t) \times \mathbf{v}(t)\|}{\|\mathbf{v}(t)\|^3}$$

**Example 13.11.** Compute the curvature  $\kappa(t)$  and the radius of curvature for the helix defined by  $\mathbf{r}(t) = \langle \cos t, \sin t, t \rangle$  for any real number  $t$ .

**Solution:** We first find the derivative of the unit tangent vector with respect to  $t$ .

```
In[124]:= Clear[r, T, t, κ]
```

```
r[t_] = {Cos[t], Sin[t], t}
r'[t]
r''[t]
```

```
Out[125]= {Cos[t], Sin[t], t}
```

```
Out[126]= {-Sin[t], Cos[t], 1}
```

```
Out[127]= {-Cos[t], -Sin[t], 0}
```

```
In[128]:= κ[t_] = Sqrt[Simplify[Cross[r''[t], r'[t]].Cross[r''[t], r'[t]]] /
           Sqrt[Simplify[r'[t].r'[t]]]^3
Out[128]= 1/2
```

It follows that  $\kappa = \frac{1}{2}$  and  $\rho = 2$  for all values of  $t$ . Hence, our helix is a curve of constant curvature.

## ■ Exercises

In Exercises 1 and 2, find  $\mathbf{r}'(t)$  and  $\mathbf{T}(t)$  and evaluate  $\mathbf{T}(2)$ .

1.  $\mathbf{r}(t) = (3 + 2t)\mathbf{i} + (2 - 5t)\mathbf{j} + 9t\mathbf{k}$       2.  $\mathbf{v}(t) = \langle \sin t, \cos t, 1 \rangle$

3. Use *Mathematica* to find the curvature function  $k(x)$  for  $y = \cos x$ . Also plot  $\kappa(x)$  for  $0 \leq x \leq 1$ . Where does the curvature assume its maximum value?

4. Determine the unit normal vectors to  $\mathbf{r}(t) = [t\mathbf{i} + \sin t\mathbf{j}]$  at  $t = \frac{\pi}{4}$  and  $t = \frac{3\pi}{4}$ .

5. Determine the curvature of the vector-valued function  $\mathbf{r}(t) = (3 + 2t)\mathbf{i} + 6t\mathbf{j} + (5 - t)\mathbf{k}$ .

6. Find a formula for the curvature of the general helix  $\mathbf{r}(t) = a \cos t\mathbf{i} + a \sin t\mathbf{j} + ct\mathbf{k}$ .

## ■ 13.5. Motion in Three Space

**Students should read Section 13.5 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.**

Recall that the velocity vector is the rate of the change of the position vector with respect to time while the acceleration vector represents the rate of change of the velocity vector with respect to time. Moreover, speed is defined to be the absolute value of the velocity vector. In short, we have the following:

$$\mathbf{v}(t) = \mathbf{r}'(t), \quad s(t) = \|\mathbf{v}(t)\| \quad \text{and} \quad \mathbf{a}(t) = \mathbf{v}'(t) = \mathbf{r}''(t)$$

One can secure the velocity vector and the position function if the acceleration vector is known via integration. More specifically:

$$\mathbf{v}(t) = \int_0^t \mathbf{a}(u) \, du + \mathbf{v}_0 \quad \text{where } \mathbf{v}_0 \text{ represents the initial velocity vector and } \mathbf{r}(t) = \int_0^t \mathbf{v}(u) \, du + \mathbf{v}_0 t + \mathbf{r}_0 \quad \text{where } \mathbf{r}_0 \text{ is the initial position.}$$

**Example 13.12.** Find the velocity vector, the speed, and the acceleration vector for the vector-valued function  $\mathbf{r}(t) = t^3\mathbf{i} + (1 - t)\mathbf{j} + 4t^2\mathbf{k}$  at time  $t = 1$ .

**Solution:**

```
In[129]:= Clear[r, v, s, a]
r[t_] = {t^3, 1 - t, 4 t^2}
v[t_] = r'[t]
s[t_] = Sqrt[v[t].v[t]]
a[t_] = r''[t]
v[1]
s[1]
a[1]
```

```
Out[130]= {t^3, 1 - t, 4 t^2}
```

```
Out[131]= {3 t^2, -1, 8 t}
```

```
Out[132]= Sqrt[1 + 64 t^2 + 9 t^4]
```

```
Out[133]= {6 t, 0, 8}
```

```
Out[134]= {3, -1, 8}
```

```
Out[135]= Sqrt[74]
```

```
Out[136]= {6, 0, 8}
```

Thus,  $\mathbf{v}(1) = \mathbf{r}'(1) = 3\mathbf{i} - \mathbf{j} + 8\mathbf{k}$ ,  $s(1) = \sqrt{74}$ , and  $\mathbf{a}(1) = 6\mathbf{i} + 8\mathbf{k}$ .

**Example 13.13.** Find  $\mathbf{r}(t)$  and  $\mathbf{v}(t)$  if  $\mathbf{a}(t) = t\mathbf{i} + 4\mathbf{j}$  subject to the initial conditions  $\mathbf{v}(0) = 3\mathbf{i} - 2\mathbf{j}$  and  $\mathbf{r}(0) = \mathbf{0}$ .

**Solution:** We first solve for  $\mathbf{v}(t)$  by integrating  $\mathbf{a}(t)$ :

```
In[137]:= Clear[r, v, a]
a[t_] = {t, 4}
v[t_] = Integrate[a[u], {u, 0, t}] + {v01, v02}
```

```
Out[138]= {t, 4}
```

```
Out[139]= {t^2/2 + v01, 4 t + v02}
```

Here, the constant vector of integration  $\mathbf{v}_0 = \langle v_{01}, v_{02} \rangle = \langle 3, -2 \rangle$  equals the initial velocity:

```
In[140]:= Solve[v[0] == {3, -2}, {v01, v02}]
```

```
Out[140]= {{v01 -> 3, v02 -> -2}}
```

Thus,  $\mathbf{v}(t) = \frac{t^2}{2}\mathbf{i} + 4t\mathbf{j} + (3\mathbf{i} - 2\mathbf{j})$ .

```
In[141]:= v[t_] = v[t] /. {v01 -> 3, v02 -> -2}
```

```
Out[141]= {3 + t^2/2, -2 + 4 t}
```

Next, we solve for  $\mathbf{r}(t)$  by integrating  $\mathbf{v}(t)$ :



```
In[142]:= r[t_] = Integrate[v[u], {u, 0, t}] + {r01, r02}
```

```
Out[142]= {r01 + 3 t +  $\frac{t^3}{6}$ , r02 - 2 t + 2 t^2}
```

Again, the constant vector of integration  $\mathbf{r}_0 = \langle r_{01}, r_{02} \rangle = \langle 0, 0 \rangle$  equals the initial position:

```
In[143]:= Solve[r[0] == {0, 0}, {r01, r02}]
```

```
Out[143]= {{r01 -> 0, r02 -> 0}}
```

Hence,  $\mathbf{r}(t) = \left(\frac{t^3}{6} + 3t\right)\mathbf{i} + (2t^2 - 2t)\mathbf{j}$ .

### Components of Acceleration

There are two components of acceleration: *tangential* and *normal*. More precisely, the acceleration vector  $\mathbf{a}$  can be decomposed as  $\mathbf{a} = a_T \mathbf{T} + a_N \mathbf{N}$ , where  $a_T = \frac{d^2 s}{dt^2} = \frac{\mathbf{a} \cdot \mathbf{v}}{\|\mathbf{v}\|}$  is the tangential component and  $a_N = \kappa \left(\frac{ds}{dt}\right)^2 = \frac{\|\mathbf{a} \times \mathbf{v}\|}{\|\mathbf{v}\|}$  is the normal component.

Moreover, one has  $a_T^2 + a_N^2 = \|\mathbf{a}\|^2$  so that  $a_N = \sqrt{\|\mathbf{a}\|^2 - a_T^2}$  and  $a_T = \sqrt{\|\mathbf{a}\|^2 - a_N^2}$ .

**Example 13.14.** Determine the tangential and normal components of acceleration for the vector function  $\mathbf{r}(t) = \langle t^3, t^2, t \rangle$ .

**Solution:**

```
In[144]:= Clear[r, v, s]
```

```
r[t_] = {t^3, t^2, t}
```

```
r'[t]
```

```
r''[t]
```

```
Out[145]= {t^3, t^2, t}
```

```
Out[146]= {3 t^2, 2 t, 1}
```

```
Out[147]= {6 t, 2, 0}
```

```
In[148]:= speed = Simplify[Sqrt[r'[t].r'[t]]]
```

```
Out[148]=  $\sqrt{1 + 4 t^2 + 9 t^4}$ 
```

The result in the last output represents the speed at time  $t$ . In order to secure the tangential component of the acceleration, we differentiate the previous output:

```
In[149]:= at = D[speed, t]
```

```
Out[149]=  $\frac{8 t + 36 t^3}{2 \sqrt{1 + 4 t^2 + 9 t^4}}$ 
```

The normal component of the acceleration is

$$\text{In[150]:= } \mathbf{an} = \sqrt{\mathbf{r}''[t] \cdot \mathbf{r}''[t] - \mathbf{at}^2}$$

$$\text{Out[150]= } \sqrt{4 + 36 t^2 - \frac{(8 t + 36 t^3)^2}{4 (1 + 4 t^2 + 9 t^4)}}$$

$$\text{In[151]:= } \mathbf{Simplify[an]}$$

$$\text{Out[151]= } 2 \sqrt{\frac{1 + 9 t^2 + 9 t^4}{1 + 4 t^2 + 9 t^4}}$$

NOTE: The components of acceleration can also be found through the formulas  $a_T = \frac{\mathbf{a} \cdot \mathbf{v}}{\|\mathbf{v}\|}$  and  $a_N = \frac{\|\mathbf{a} \times \mathbf{v}\|}{\|\mathbf{v}\|}$ , confirmed using *Mathematica* as follows:

$$\text{In[152]:= } \mathbf{at} = \mathbf{r}''[t] \cdot \mathbf{r}'[t] / \text{Sqrt}[\mathbf{r}'[t] \cdot \mathbf{r}'[t]]$$

$$\mathbf{an} = \text{Sqrt}[\text{Cross}[\mathbf{r}''[t], \mathbf{r}'[t]] \cdot \text{Cross}[\mathbf{r}''[t], \mathbf{r}'[t]]] / \text{Sqrt}[\mathbf{r}'[t] \cdot \mathbf{r}'[t]]$$

$$\text{Out[152]= } \frac{4 t + 18 t^3}{\sqrt{1 + 4 t^2 + 9 t^4}}$$

$$\text{Out[153]= } \frac{\sqrt{4 + 36 t^2 + 36 t^4}}{\sqrt{1 + 4 t^2 + 9 t^4}}$$

## ■ Exercises

In Exercises 1 and 2, calculate the velocity and acceleration vectors and the speed at the time indicated:

1.  $\mathbf{r}(t) = t^2 \mathbf{i} + (1 - t) \mathbf{j} + (5 t^2) \mathbf{k}$ ,  $t = 2$ .

2.  $\mathbf{r}(t) = \cos t \mathbf{i} + \sin t \mathbf{j} + \tan(2t) \mathbf{k}$ ,  $t = \frac{\pi}{6}$ .

3. Sketch the path  $\mathbf{r}(t) = (1 - t^2) \mathbf{i} + (1 - t) \mathbf{j}$  for  $-3 \leq t \leq 3$  and compute the velocity and acceleration vectors at  $t = 0$ ,  $t = 1$ , and  $t = 2$ .

4. Find  $\mathbf{v}(t)$  given  $\mathbf{a}(t)$  and the initial velocity  $\mathbf{v}_0$ .

a.  $\mathbf{a}(t) = t \mathbf{i} + 3 \mathbf{j}$ ,  $\mathbf{v}(0) = \frac{1}{2} \mathbf{i} + 2 \mathbf{j}$

b.  $\mathbf{a}(t) = e^{2t} \mathbf{i} + 0 \mathbf{j} + (t + 2) \mathbf{k}$ ,  $\mathbf{v}(0) = \mathbf{i} - 3 \mathbf{j} + 2 \mathbf{k}$

5. Find  $\mathbf{r}(t)$  and  $\mathbf{v}(t)$  given  $\mathbf{a}(t)$  together with the initial velocity and position at rest:

a.  $\mathbf{a}(t) = e^{3t} \mathbf{i} + 4t \mathbf{j} + (t - 2) \mathbf{k}$ ,  $\mathbf{v}(0) = 0 \mathbf{i} + 0 \mathbf{j} + \mathbf{k}$ ,  $\mathbf{r}(0) = 0 \mathbf{i} + 3 \mathbf{j} + 4 \mathbf{k}$

b.  $\mathbf{a}(t) = 0 \mathbf{i} + 0 \mathbf{j} + \sin t \mathbf{k}$ ,  $\mathbf{v}(0) = \mathbf{i} + \mathbf{j}$ ,  $\mathbf{r}(0) = \mathbf{i}$ .

6. Find the decomposition of  $\mathbf{a}(t)$  into its tangential and normal components at the indicated point:

a.  $\mathbf{r}(t) = (3 - 4t) \mathbf{i} + (t + 1) \mathbf{j} + t^2 \mathbf{k}$  at  $t = 1$

b.  $\mathbf{r}(t) = t \mathbf{i} + e^{-t} \mathbf{j} + t e^{-t} \mathbf{k}$  at  $t = 0$

7. Show that the tangential and normal components of acceleration of the helix given by  $\mathbf{r}(t) = (\cos t) \mathbf{i} + (\sin t) \mathbf{j} + t \mathbf{k}$  are equal to 0 and 1, respectively.